



HAL
open science

Un programme BASIC d'estimation de paramètres de deux populations animales sujettes à migrations entre deux zones

Frédéric Lardeux, Stéphane Loncke

► **To cite this version:**

Frédéric Lardeux, Stéphane Loncke. Un programme BASIC d'estimation de paramètres de deux populations animales sujettes à migrations entre deux zones. Rapport ITRMLM n°14/87/ITRM/Doc.Ent, Institut Louis Malardé, Tahiti, French Polynesia. 1987, pp.1-25. hal-04701272

HAL Id: hal-04701272

<https://hal.science/hal-04701272v1>

Submitted on 18 Sep 2024

HAL is a multi-disciplinary open access archive for the deposit and dissemination of scientific research documents, whether they are published or not. The documents may come from teaching and research institutions in France or abroad, or from public or private research centers.

L'archive ouverte pluridisciplinaire **HAL**, est destinée au dépôt et à la diffusion de documents scientifiques de niveau recherche, publiés ou non, émanant des établissements d'enseignement et de recherche français ou étrangers, des laboratoires publics ou privés.



Distributed under a Creative Commons Attribution 4.0 International License

I.T.R.L.M.
B.P. 30
Papeete - Tahiti

O.R.S.T.O.M.
B.P. 529
Papeete - Tahiti

UN PROGRAMME BASIC D'ESTIMATION DE PARAMETRES
DE DEUX POPULATIONS ANIMALES SUJETTES A
MIGRATIONS ENTRE DEUX ZONES

F. LARDEUX
S. LONCKE

Décembre 1987

Référence ITRMLM : 14 / 87 / ITRM / Doc-Ent.

PLAN

INTRODUCTION

1 - LE MODELE

- 1.1 - Expérimentation - notations
- 1.2 - Modèle général
 - 12.1 - Hypothèses
 - 12.2 - Paramètres
- 1.3 - Sous-modèle à migrations partielles (modèle d'Iwao)
 - 13.1 - Hypothèses
 - 13.2 - Paramètres
- 1.4 - Sous-modèle à migrations restreintes
 - 14.1 - Hypothèses
 - 14.2 - Paramètres
- 1.5 - Comparaison entre les modèles - précision des estimations

2 - LE PROGRAMME

- 2.1 - Description générale
- 2.2 - Opérations sur les fichiers
 - 22.1 - création d'un fichier
 - 22.2 - listage d'un fichier
 - 22.3 - corrections dans un fichier
- 2.3 - Calcul des paramètres des modèles
- 2.4 - Mises en garde

CONCLUSION

ANNEXE 1 : Listing du programme

INTRODUCTION

Le programme présenté ici est l'application directe des travaux d'ARNASON (1972).

Le programme permet d'estimer certains paramètres démographiques et comportementaux de deux populations sujettes à migrations (et mortalité) entre deux zones. Il a été développé pour estimer de tels paramètres (et en particulier le "taux d'échange") de deux populations de moustiques adultes (Aedes polynesiensis) pouvant migrer entre deux îlots (d'un atoll de Polynésie française) distants d'une cinquantaine de mètres (LARDEUX, 1987).

1 - LE MODELE

On reprend ici l'analyse théorique du modèle développé par ARNASON (1972), qui a amélioré les estimateurs formulés par RICHARDS et WALOFF (1954) puis par IWAO (1963), dans le cadre d'un échantillonnage en trois occasions.

1.1 - Expérimentation - notations.

Deux zones (notées A et B) sont échantillonnées à trois reprises chacune.

Premier échantillonnage : les animaux sont comptés, marqués et relâchés dans leur zone d'origine.

Deuxième et troisième échantillonnage : les animaux marqués sont comptés et répertoriés selon leur zone de capture. Les animaux non marqués sont comptés et marqués. Tout l'échantillon est alors relâché dans sa zone de capture.

Les deux zones sont échantillonnées selon ce protocole.

La notation utilisée est celle d'ARNASON. Elle a aussi été conservée pour le nom des variables dans le programme informatique. On a :

$j = 1, 2, 3$		indique le moment de l'échantillonnage
$i = 1, \dots, (j-1)$		
$x = A \text{ ou } B$		indique la zone d'échantillonnage
$y = A \text{ ou } B$		

Observations : (c.à d. ce que l'on obtient sur le terrain)

m_{ijxy} = nombre d'animaux capturés dans l'échantillon j de la zone y qui ont avant été capturés dans l'échantillon i de la zone x

u_{jy} = nombre d'animaux non marqués capturés dans l'échantillon j de la zone y

n_{jy} = taille de l'échantillon j de la zone y ($= m_{ijxy} + u_{jy}$)

Paramètres :

\emptyset_{jxy} = probabilité qu'un animal, vivant et dans la zone x au temps j soit vivant et dans la zone y au temps $j+1$

\emptyset_{jx} = probabilité qu'un animal, vivant et dans la zone x au temps J , soit vivant au temps $J+1$ ($= \emptyset_{jxA} + \emptyset_{jxB}$)

p_{jx} = probabilité de capture d'un animal dans la zone x , au temps j

Variables aléatoires non observables :

M_{ijxy} = nombre d'animaux, vivants et dans la zone y au moment où le j ème échantillon est pris, qui ont été auparavant capturés au temps i dans la zone x

U_{jy} = nombre d'animaux non marqués vivant dans la zone y au moment du j ème échantillon

N_{jy} = taille de la population vivant dans la zone y au temps j

Le modèle d'ARNASON permet d'estimer les \emptyset_{jxy} , p_{jx} , N_{jy} . On remarque que dans ce sens, il est totalement stochastique puisqu'il estime des probabilités.

Selon les hypothèses de départ (notamment sur les modalités des migrations possibles), il discerne deux sous-modèles issus d'un modèle général.

1.2 - Modèle général.

12.1 - Hypothèses.

Elles sont les suivantes :

- 1 - tous les animaux vivants et dans la zone x au temps i ont la même probabilité de capture dans le i ème échantillon.

- 2 - tous les animaux vivants et dans la zone x après que le i ème échantillon ait été pris, ont la même probabilité de survie jusqu'au i+1 ème échantillonnage (indifféramment de leurs possibilités de migrations).
- 3 - les animaux peuvent migrer librement entre les deux zones A et B . Si ils émigrent au delà de ces zones ils le font de façon permanente.
- 4 - tous les animaux de la zone x au temps i ont la même probabilité conditionnelle de se trouver dans l'autre zone au temps i+1, sachant qu'ils survivent jusqu'à i+1. Les animaux sont libres de migrer plus d'une fois entre deux périodes d'échantillonnage, ou au cours de l'échantillonnage. En particulier, cette hypothèse reste vraie même si tous les animaux ont la même tendance de migration. Elle sera faussée si par exemple, une partie de la population est sédentaire durant l'expérimentation. La population doit donc avoir un "comportement homogène".
- 5 - il ne doit pas y avoir de "pertes de marques".
- 6 - il ne doit pas y avoir d'immigration dans les deux zones durant l'expérimentation.

Les probabilités de survie, capture et migration ne doivent pas être affectées par l'âge des animaux, leurs captures antérieures, leur taille, la densité de la population etc... mais peuvent être différentes entre les zones et les périodes d'échantillonnage.

12.2 - Paramètres.

Les estimations sont données pour la zone A seulement. En raison de la symétrie de l'expérimentation, les estimations pour la zone B sont obtenues en interchangeant les indices A et B dans les équations.

On se reportera à la publication d'ARNASON pour le détail des calculs. Les résultats sont :

soit $DET(i, j) = m_{ijAA} \cdot m_{ijBB} - m_{ijBA} \cdot m_{ijAB}$
on a alors :

$$p_{1A} = DET(1, 2) / (n_{2A} \cdot m_{12BB} - n_{2B} \cdot m_{12BA})$$

$$N_{1A} = n_{1A} / p_{1A}$$

$$\phi_{1AA} = (m_{12AA} / n_{1A}) + n_{2A} \cdot (m_{13AA} \cdot m_{23BB} - m_{13AB} \cdot m_{23BA}) / (n_{1A} \cdot DET(2, 3))$$

$$\phi_{1AB} = (m_{12AB} / n_{1A}) + n_{2B} \cdot (m_{13AB} \cdot m_{23AA} - m_{13AA} \cdot m_{23AB}) / (n_{1A} \cdot DET(2, 3))$$

$$\phi_{1A} = \phi_{1AA} + \phi_{1AB}$$

$$p2A = \text{DET}(2,3) / (n3A \cdot m23BB - n3B \cdot m23BA)$$

$$N2A = n2A / p2A$$

$$\emptyset2AA.p3A = m23AA / n2A$$

$$\emptyset2BA.p3A = m23BA / n2B$$

1.3 - Sous-modèle à migrations partielles (modèle d'Iwao).

13.1 - Hypothèses.

Les hypothèses sous-jacentes à ce modèle sont les mêmes que celles du modèle général en ce qui concerne les points 1, 2, 3, 5 et 6. Le point 4 doit être modifié de la façon suivante :

- 4 - tous les animaux dans la zone x au temps 1 ont la même probabilité conditionnelle d'être dans l'autre zone au temps 2, sachant qu'ils survivent jusqu'au temps 2. S'ils migrent dans la zone opposée, ils deviennent alors sédentaires et ne remigrent plus.

Aussi, si des animaux sont capturés avec le code m123ABA ou m123BAB il est évident que l'hypothèse précédente n'est plus respectée et que les équations du modèle général doivent être utilisées. L'inverse n'est pas vrai : l'absence d'observations m123ABA ou m123BAB n'implique pas que le modèle général n'est pas approprié. Il y a une forte probabilité, si on utilise le modèle général, qu'aucune observation de ce type n'apparaisse, spécialement si l'effort d'échantillonnage est faible ou si les populations sur chaque zone sont petites ou possèdent un faible taux de survie.

13.2 - Paramètres.

Les estimations sont les mêmes que pour le modèle général sauf pour les paramètres $\emptyset1AA$ et $\emptyset1AB$ et $\emptyset2BA.p3A$ (Pour la zone B, intervertir les indices A et B). On a donc

$$\emptyset1AB = \emptyset1BB \cdot n1B \cdot m12AB / (n1A \cdot m12AA)$$

$$\text{avec, } \emptyset1BB = (n2B \cdot m13BB + m12BB \cdot M23BB) / (m23BB \cdot n1B)$$

$$\emptyset2BA.p3A = m23BA \cdot p1B / m12BB$$

1.4 - Sous-modèle à migrations restreintes.

14.1 - Hypothèses.

Si l'hypothèse que certains animaux, une fois leur migration effectuée, deviennent "sédentaires" pour une durée supérieure à celle de l'expérimentation, il est normal de considérer qu'une partie de la population était déjà sédentarisée au temps 1. Ce

modèle tient compte de ce fait. On construit alors des groupes d'animaux tels que :

$$N1A = N1A(g) + N1A(s)$$

où N1A(s) sont les "sédentaires" et N1A(g) la classe des animaux au comportement "général", au temps 1

De même, $N2A = N2A(g) + N2A(s)$

Le point d'hypothèse n° 4 devient (les autres hypothèses, 1, 2, 3, 5 et 6 étant maintenues):

- 4 - il existe une partie de la population de la zone x au temps 1 qui est "sédentaire" (les N1x(s)). Ces animaux peuvent survivre au temps 2 (cf. point 2) dans la zone x. Dans ce cas, ils rejoignent le groupe des N2x(s). Les N1x(g) peuvent survivre au temps 2, rejoignant alors soit les N2x(g), soit les N2y(s) (x ≠ y).

14.2 - Paramètres.

les estimations sont les mêmes que pour le modèle précédent (modèle d'IWAD) sauf que les estimations en zone A de ø1AB et ø2BA.p3A ne sont pas calculables (même chose pour la zone B).

1.5 - Comparaison entre les modèles - précision des estimations.

D'après ARNASON, les estimations des paramètres sont asymptotiquement non biaisées. Quelque soit le modèle choisi, ces propriétés restent valables et dans ce sens, certaines estimations sont "robustes" aux contraintes introduites par les deux sous-modèles, comme le montre le tableau suivant (un signe égal dans la colonne indique que l'estimation est valable quelque soit le modèle utilisé) :

modèle général		modèle "partiel"		modèle "restreint"
p1A	=	p1A	=	p1A
N1A	=	N1A	=	N1A
ø1AA	#	ø1AA	=	ø1AA
ø1AB	#	ø1AB		non calculable
N2A	=	N2A	=	N2A
p2A	=	p2A	=	p2A
ø2AA.p3A	=	ø2AA.p3A	=	ø2AA.p3A
ø2BA.p3A	=	ø2BA.p3A		non calculable
N3A.p3A	=	N3A.p3A	=	N3A.p3A

Les formules de variance des estimateurs n'ont pas été développées par ARNASON. Il précise cependant que les variances sont d'autant plus faibles que les contraintes des modèles sont fortes (ce qui est le cas dans les deux sous-modèles). Il signale par exemple un gain de 50 % dans la précision des variances des θ_{ixy} et π_{ix} lorsqu'on passe du modèle général au modèle d'Iwao.

Malheureusement, les deux-sous modèles du modèle général ont des hypothèses non réalistes en pratique (et certains paramètres non estimables dans le cas du dernier modèle). C'est pourquoi, ARNASON ne donne des remarques sur la précision des estimations que pour le modèle général.

La qualité des estimations (précision, biais, fréquence d'occurrence de valeurs erronées ...) n'est pas seulement affectée par l'intensité de l'échantillonnage mais aussi par la taille de la population échantillonnée (elle même affectée par les probabilités de survie, de migrations, donc par le temps séparant les périodes d'échantillonnage).

Par simulation, ARNASON a donné deux règles empiriques :

- si toutes les valeurs observées m_{ixy} sont plus grandes que 5, alors les estimations peuvent être considérées comme non biaisées et raisonnablement précises (écart-type inférieur à la moitié de la valeur de l'estimation du paramètre). Des valeurs erronées ont peu de chance de se produire.
- si l'expérimentateur a une idée à priori de la valeur des paramètres N_{ix} et θ_{ixy} , il peut calculer, pour planifier son expérience, les valeurs m_{ixy} , qui de toute façon, doivent être supérieures à 8 (et si possible à 10). Alors, si les valeurs à priori sont plus ou moins correctes, la précision des estimations sera bonne (écart-type inférieur à la moitié de la valeur de l'estimation), le biais négligeable et la probabilité de calcul de valeurs erronées faible (< 0.2)

En général, si les valeurs de N_{1A} et N_{1B} sont inférieures à 200, l'expérimentation ne vaut pas la peine d'être effectuée, à moins que la quasi-totalité de la population ne soit capturée au cours du premier échantillonnage.

2 - LE PROGRAMME

2.1 - Description générale.

Le programme proposé ici, écrit en GWBASIC sur un compatible IBM-PC (Goupil G4), permet :

- de créer des fichiers spécifiques au calcul des estimations des paramètres des modèles ci-dessus.
- de modifier (corriger) ces fichiers en cas d'erreur de saisie.
- de lister (à l'écran ou sur imprimante) ces fichiers.
- de calculer les estimations des paramètres du modèle général et des deux sous-modèles (modèle à migration partielle et modèle à migration restreinte).

Le programme a été écrit dans l'optique d'une utilisation la plus conviviale possible : on accède aux diverses options à l'aide de menus. Aucune connaissance en programmation n'est nécessaire pour faire tourner ce programme.

Mise en route du programme.

Il suffit de charger le GWBASIC puis de taper RUN "\$\$\ARNASON"

\$\$ est le chemin d'accès au programme

ARNASON est le nom du programme

Par exemple : RUN "C:\DYN\ARNASON" si le programme ARNASON se trouve sur un disque dur, dans le répertoire DYN
ou simplement RUN "ARNASON" si le programme se trouve dans le même répertoire (ou sur la même disquette) que GWBASIC.

Ensuite, il suffit de suivre les indications portées à l'écran : un premier écran apparaît, indiquant la référence de la publication source (figure 1). Faites un 'retour-chariot' pour accéder au menu principal.

Le menu principal permet de travailler sur les fichiers spécifiques à ce programme, de calculer les estimations des paramètres des modèles ou de terminer la session de travail (fig. 2).

Choisissez simplement le numéro correspondant à l'option désirée et faites un 'retour-chariot'.

2.2 - Opérations sur les fichiers

22.1 - Création de fichier.

Lorsqu'on désire créer un fichier, le programme demande tout d'abord le nom du fichier dans lequel seront stockées les données, avec son chemin d'accès complet. Par exemple, si on désire créer un fichier portant le nom EXP1, dans le répertoire DONNEES du disque dur, il faut entrer la séquence suivante : C:\DONNEES\EXP1

Le programme ajoute automatiquement l'extension .MIG (pour 'migration') au nom du fichier, ce qui permet de repérer facilement

```

*****
* ESTIMATION DES PARAMETRES DE POPULATIONS *
* SUJETTES A DES MIGRATIONS ENTRE DEUX ZONES *
*****

```

SOURCE : ARNASON A.N. - 1972 - Parameters estimates from mark-recapture experiments on two populations subject to migration and death. Res. Popul. Ecol. XIII : 97-113

PROGRAMME : F. LARDEUX et S. LONCKE
 Centre ORSTOM - BP 529
 Papeete . Tahiti

Décembre 1987

Faites un 'retour-chariot 'pour continuer

Fig. 1 : Ecran de présentation

```

*****
* ESTIMATION DES PARAMETRES DE POPULATIONS *
* SUJETTES A DES MIGRATIONS ENTRE DEUX ZONES *
*****

```

```

CREATION D'UN FICHER ..... 1
LISTAGE D'UN FICHER ..... 2
CORRECTIONS DANS UN FICHER ... 3
ESTIMATION DES PARAMETRES ..... 4

FIN DU PROGRAMME ..... 5

```

NUMERO CHOISI :

Fig. 2 : Menu général

sur le disque tous les fichiers créés par le programme. Ainsi, le fichier EXP1 est inscrit sous le nom EXP1.MIG dans le répertoire DONNEES du disque dur C.

Par la suite, lorsque le programme demande un nom de fichier (listage, corrections, calcul des estimations ...), il n'est pas nécessaire de donner l'extension .MIG. Seul le nom sans extension doit être donné. Si l'extension .MIG est donnée par l'utilisateur, il se produit un message d'erreur et on doit relancer le programme.

La figure 3 donne un exemple des questions posées par le programme. Les données sont rentrées dans l'ordre de leur obtention sur le terrain : premier échantillonnage (nombre d'animaux marqués dans la zone A puis dans la zone B), deuxième échantillonnage, puis troisième échantillonnage, avec à chaque fois la zone A et B séparées. Il suffit donc de suivre les indications données à l'écran. L'exemple de la figure 3 est bâti sur les valeurs données par ARNASON dans sa publication.

Vu le faible nombre de données à saisir (18), il n'a pas été prévu de système de correction au moment de la saisie. Une erreur pourra cependant être corrigée avec l'option "CORRECTION DANS UN FICHIER" du menu principal (cf. paragraphe 22.3). L'utilisateur doit donc d'abord finir de créer son fichier avant d'utiliser cette option.

Une fois le fichier créé, le programme réaffiche le menu principal.

22.2 - Listage d'un fichier.

Comme précédemment, le programme demande le nom du fichier que l'on désire lister. Si l'utilisateur travaille déjà sur un fichier, le programme considère le fichier courant comme fichier à lister : la question posée est "EST CE VOTRE FICHIER DE TRAVAIL (O/N) : ". Répondre O (= oui) ou N (= non). Dans le dernier cas, l'utilisateur peut changer de fichier de travail. Il suffit de répondre à la question apparaissant à l'écran : "NOM DU FICHIER (avec le chemin d'accès complet) : ". On rappelle qu'un nom de fichier ne doit pas être rentré avec son extension .MIG : le programme se charge de la rajouter.

Le listage du fichier se fait à l'écran. Ensuite, on peut sortir ce qui est apparu à l'écran, au niveau de l'imprimante. Il suffit de répondre O à la question : "IMPRESSION SUR IMPRIMANTE (O/N) : ". Un Exemple de sortie est données à la figure 4.

Une fois le listage terminé, le programme réaffiche le menu général.

22.3 - Correction dans un fichier.

La procédure de demande de nom du fichier à corriger est la même que précédemment.

CREATION DE FICHER

NOM DU FICHER : C:\DONNEES\EXP1.MIG

PREMIER ECHANTILLONNAGE (Temps t = 1)

TOTAL CAPTURE ET MARQUE DANS LA ZONE A : 193
TOTAL CAPTURE ET MARQUE DANS LA ZONE B : 228

DEUXIEME ECHANTILLONNAGE (Temps t = 2)

ZONE A :

TOTAL CAPTURE EN A : 85
NBRE MARQUES EN A (à t = 1) ET RECAPTURES EN A : 31
NBRE MARQUES EN B (à t = 1) ET RECAPTURES EN A : 7

ZONE B :

TOTAL CAPTURE EN B : 176
NBRE MARQUES EN A (à t = 1) ET RECAPTURES EN B : 8
NBRE MARQUES EN B (à t = 1) ET RECAPTURES EN B : 64

TROISIEME ECHANTILLONNAGE

ZONE A :

TOTAL CAPTURE EN A : 84
NBRE MARQUES EN A (à t = 1) ET RECAPTURES EN A : 9
NBRE MARQUES EN A (à t = 2) ET RECAPTURES EN A : 18
NBRE MARQUES EN B (à t = 1) ET RECAPTURES EN A : 12
NBRE MARQUES EN B (à t = 2) ET RECAPTURES EN A : 11

ZONE B :

TOTAL CAPTURE EN B : 144
NBRE MARQUES EN A (à t = 1) ET RECAPTURES EN B : 9
NBRE MARQUES EN A (à t = 2) ET RECAPTURES EN B : 9
NBRE MARQUES EN B (à t = 1) ET RECAPTURES EN B : 33
NBRE MARQUES EN B (à t = 2) ET RECAPTURES EN B : 42

Fig. 3 : Exemple de création de fichier

```

+++++
+   PARAMETRES DE DEUX POPULATIONS   +
+ SUJETTES A MIGRATIONS ENTRE 2 ZONES +
+++++

```

NOM DU FICHIER : C:\DONNEES\EXP1.MIG

PREMIER ECHANTILLONNAGE (Temps t = 1)

TOTAL CAPTURE ET MARQUE DANS LA ZONE A : 193
TOTAL CAPTURE ET MARQUE DANS LA ZONE B : 228

DEUXIEME ECHANTILLONNAGE (Temps t = 2)

ZONE A :

TOTAL CAPTURE EN A : 85
NBRE MARQUES EN A (à t = 1) ET RECAPTURES EN A : 31
NBRE MARQUES EN B (à t = 1) ET RECAPTURES EN A : 7

ZONE B :

TOTAL CAPTURE EN B : 176
NBRE MARQUES EN A (à t = 1) ET RECAPTURES EN B : 8
NBRE MARQUES EN B (à t = 1) ET RECAPTURES EN B : 64

TROISIEME ECHANTILLONNAGE

ZONE A :

TOTAL CAPTURE EN A : 84
NBRE MARQUES EN A (à t = 1) ET RECAPTURES EN A : 9
NBRE MARQUES EN A (à t = 2) ET RECAPTURES EN A : 18
NBRE MARQUES EN B (à t = 1) ET RECAPTURES EN A : 12
NBRE MARQUES EN B (à t = 2) ET RECAPTURES EN A : 11

ZONE B :

TOTAL CAPTURE EN B : 144
NBRE MARQUES EN A (à t = 1) ET RECAPTURES EN B : 9
NBRE MARQUES EN A (à t = 2) ET RECAPTURES EN B : 9
NBRE MARQUES EN B (à t = 1) ET RECAPTURES EN B : 33
NBRE MARQUES EN B (à t = 2) ET RECAPTURES EN B : 42

Fig. 4 : Exemple de listage d'un fichier

Le programme affiche alors à l'écran les valeurs des données, selon la notation utilisée dans la présentation théorique du modèle (figure 5). Par exemple, M12AB signifie : nombre d'animaux marqués en A au temps 1 (1^{er} échantillonnage) et capturés en B au temps 2 (2^{ème} échantillonnage); N2B signifie : nombre total d'animaux capturés (marqués et non marqués) dans la zone B, au temps 2.

Chaque donnée est affectée d'un numéro. Si une valeur est à modifier, il faut donner le numéro correspondant, à la question apparaissant à l'écran. Le programme affiche alors la valeur en cours et demande la nouvelle valeur. Une fois celle-ci saisie, le programme réaffiche l'ensemble des données (avec les corrections) et redemande si d'autres données sont à corriger. Dans la négative, faire un 'retour-chariot' (ou entrer zéro) pour arrêter le processus. Le programme réaffiche alors le menu général.

2.3 - Calcul des paramètres des modèles.

Lorsque cette option est choisie, le programme affiche un deuxième menu où on peut choisir entre les trois modèles décrits précédemment : modèle général, sous-modèle à migration partielle, sous-modèle à migration restreinte; et une option de retour au menu principal (fig. 6)

La procédure de demande de nom de fichier est la même que précédemment.

Une fois le modèle choisi, les estimations apparaissent à l'écran. Comme pour le listage du fichier, une option "IMPRESSION SUR IMPRIMANTE" est prévue. Un exemple de sortie est donné à la figure 7. Si l'utilisateur ne désire pas de sortie sur l'imprimante, le programme réaffiche le menu de choix du modèle.

2.4 - Mises en garde.

- 1 - Lors de la création d'un fichier, bien vérifier que le nom donné n'existe pas déjà. Aucune vérification n'a été prévue dans le programme et le nouveau fichier détruira l'ancien en cas de conflit nominal.
- 2 - Ne pas oublier de ne jamais mettre d'extension au nom de fichier. Le programme se charge de rajouter l'extension '.MIG' à tous les fichiers qu'il peut traiter.

CONCLUSION

Le programme présenté ici n'est peut-être pas optimal. Il permet cependant de traiter rapidement des données de terrain pour estimer des paramètres de populations sujettes à migrations entre deux zones, selon les modèles présentés par ARNASON.

FICHER : EXP1.MIG

N1A	=	193	1
N1B	=	228	2
N2A	=	85	3
M12AA	=	31	4
M12BA	=	7	5
N2B	=	176	6
M12AB	=	8	7
M12BB	=	64	8
N3A	=	84	9
M13AA	=	9	10
M23AA	=	18	11
M13BA	=	12	12
M23BA	=	11	13
N3B	=	144	14
M13AB	=	9	15
M23AB	=	9	16
M13BB	=	33	17
M23BB	=	42	18

NUMERO A CORRIGER (0 = arrêt des corrections) :

Fig. 5: Exemple d'écran de correction de fichier

CHOIX DU MODELE

MODELE GENERAL	1
MODELE A MIGRATION PARTIELLE (MODELE D'IWAO)	2
MODELE A MIGRATION RESTREINTE	3
RETOUR AU MENU GENERAL	4

NUMERO CHOISI :

Fig. 6 : Menu de calcul des estimations

On peut contacter les auteurs de ce rapport pour tout renseignement complémentaire ou pour la fourniture sur disquette du programme.

```

+++++
+   PARAMETRES DE DEUX POPULATIONS   +
+ SUJETTES A MIGRATIONS ENTRE 2 ZONES +
+++++

```

FICHER : C:\DONNEES\EXP1.MIG

MODELE GENERAL

PHI(j,x,y) = probabilité qu'un animal, vivant et dans la zone x au temps j, soit vivant et dans la zone y au temps j+1

PHI(1,A,A) = .3476471	PHI(1,A,B) = .1538789
PHI(1,B,B) = .8517184	PHI(1,B,A) = .1107106

PHI(j,x) = probabilité qu'un animal, vivant et dans la zone x au temps j, soit vivant au temps j+1 [= PHI(j,x,A) + PHI(j,x,B)]

PHI(1,A) = .501526	PHI(1,B) = .9624289
----------------------	-----------------------

P(j,x) = probabilité qu'un animal, vivant et dans la zone x au temps j, soit capturé dans l'échantillon pris au temps j.

P(1,A) = .4581749	P(1,B) = .4036851
P(2,A) = .337963	P(2,B) = .3578431

N(j,x) = taille de la population vivant dans la zone x au temps j

N(1,A) = 421.2365	N(1,B) = 564.7967
N(2,A) = 251.5069	N(2,B) = 491.8357

PHI(2,A,A)*P(3,A) = .2117647	PHI(2,B,A)*P(3,A) = .0625
PHI(2,B,B)*P(3,B) = .2386364	PHI(2,A,B)*P(3,B) = .1058824

Fig. 7 : Exemple d'estimation des paramètres du "modèle général"

BIBLIOGRAPHIE

- ARNASON A.N. - 1972 - Parameter estimates from mark-recapture experiments on two populations subject to migration and death. Res. Popul. Ecol. XIII : 97-113
- IWAO S. - 1963 - On a method for estimating the rate of population interchange between two areas. Res. Popul. Ecol. 5 : 44-50
- LARDEUX F. - 1987 - Lutte biologique contre Aedes polynesiensis avec le copépode Mesocyclops aspericornis. Expérimentation à Rangiroa. I - Etude avant traitement - Traitement. Rapport ITRMLM n° 13/87/ITRM/DOC-ENT.
- RICHARDS O. W., WALOFF N. - 1954 - Studies on the biology and population dynamics of british grasshoppers. Anti-Locust. Bull. 17 : 1-82

ANNEXE 1 : LISTING DU PROGRAMME

```

10 /+++++
20 /
30 /          PROGRAMME ARNASON
40 /
50 /          Estimation des paramètres d'expérimentations de capture
60 /          recapture sur deux populations sujettes à migrations et
70 /          mortalité.
80 /
90 /          source : ARNASON - 1972 - Res. Popul. Ecol. XIII : 97-113
100 /          Programme Frédéric LARDEUX . ORSTOM-TAHITI BP 529, Papeete
110 /          Décembre 1987
120 /
130 /+++++
140 /
150 CLEAR
160 KEY OFF
170 DIM A(18)
180 /
190 CLS
200 PRINT :PRINT :PRINT
210 PRINT TAB(18)"*****"
220 PRINT TAB(18)"* ESTIMATION DES PARAMETRES DE POPULATIONS *"
230 PRINT TAB(18)"* SUJETTES A DES MIGRATIONS ENTRE DEUX ZONES *"
240 PRINT TAB(18)"*****"
250 PRINT :PRINT:PRINT
260 PRINT "          SOURCE      : ARNASON A.N.- 1972 - Parameters estimates from
          mark-recapture experiments on two populations
          subject to migration and death.
          Res. Popul. Ecol. XIII : 97-113"
270 PRINT "
280 PRINT :PRINT
290 PRINT "          PROGRAMME : F. LARDEUX et S. LONCKE "
300 PRINT "          Centre ORSTOM - BP 529
          Papeete . Tahiti"
310 PRINT
320 PRINT "          Décembre 1987
330 PRINT :PRINT
340 INPUT "          Faites un 'retour-chariot 'pour continuer .... ",B#
350 CLS
360 PRINT :PRINT :PRINT
370 PRINT TAB(18)"*****"
380 PRINT TAB(18)"* ESTIMATION DES PARAMETRES DE POPULATIONS *"
390 PRINT TAB(18)"* SUJETTES A DES MIGRATIONS ENTRE DEUX ZONES *"
400 PRINT TAB(18)"*****"
410 PRINT :PRINT:PRINT
420 PRINT TAB(25) "CREATION D'UN FICHER ..... 1"
430 PRINT TAB(25) "LISTAGE D'UN FICHER ..... 2"
440 PRINT TAB(25) "CORRECTIONS DANS UN FICHER ... 3"
450 PRINT TAB(25) "ESTIMATION DES PARAMETRES ..... 4"
460 PRINT
470 PRINT TAB(25) "FIN DU PROGRAMME ..... 5"
480 PRINT:PRINT :PRINT
490 PRINT TAB(30) "NUMERO CHOISI : ";:INPUT "",CREAT
500 IF CREAT<1 OR CREAT>5 THEN 490

```

```

510 IF CREAT=5 THEN END
520 GOSUB 1970
530 ON CREAT GOTO 550,700,840,1130,540
540 END
550 '
560 '+++++++ CREATION DE FICHER ++++++
570 '
580 CLS
590 PRINT :PRINT :PRINT TAB(30) "CREATION DE FICHER "
600 CREAT = 1
610 A$ = "N"
620 K = 1
630 GOSUB 2560
640 CLOSE :OPEN "O",#1,F$
650 FOR I = 1 TO 18 : PRINT #1,A(I) : NEXT I : CLOSE #1
660 GOTO 350
670 '
680 '*****
690 '
700 '+++++++ LISTAGE D'UN FICHER ++++++
710 '
720 K = 1
730 CLOSE :OPEN "I",#1,F$
740 FOR I = 1 TO 18 : INPUT #1,A(I) : NEXT I
750 CLOSE 1
760 CLS
770 A$="N"
780 GOSUB 2560
790 PRINT :PRINT :INPUT "IMPRESSION SUR L'IMPRIMANTE ? (O/N) : ",A$
800 IF (A$<>"O" AND A$<>"N" AND A$<>"o" AND A$<>"n" ) THEN 790
810 IF A$="o" OR A$="O" THEN CLOSE #1 : K = 1 : GOSUB 2560
820 GOTO 350
830 '
840 '+++++++ CORRECTIONS DANS UN FICHER ++++++
850 CLOSE :OPEN "I",#1,F$
860 FOR I= 1 TO 18 : INPUT #1,A(I):NEXT I
870 CLOSE 1
880 CLS
890 PRINT :PRINT "FICHER : ";F$:PRINT
900 PRINT "          N1A   = ";A(1),:PRINT "          1"
910 PRINT "          N1B   = ";A(2),:PRINT "          2"
920 PRINT "          N2A   = ";A(3),:PRINT "          3"
930 PRINT "          M12AA = ";A(4),:PRINT "          4"
940 PRINT "          M12BA = ";A(5),:PRINT "          5"
950 PRINT "          N2B   = ";A(6),:PRINT "          6"
960 PRINT "          M12AB = ";A(7),:PRINT "          7"
970 PRINT "          M12BB = ";A(8),:PRINT "          8"
980 PRINT "          N3A   = ";A(9),:PRINT "          9"
990 PRINT "          M13AA = ";A(10),:PRINT "         10"
1000 PRINT "          M23AA = ";A(11),:PRINT "         11"

```

```

1010 PRINT "          M13BA = ";A(12),:PRINT " ..... 12"
1020 PRINT "          M23BA = ";A(13),:PRINT " ..... 13"
1030 PRINT "          N3B   = ";A(14),:PRINT " ..... 14"
1040 PRINT "          M13AB = ";A(15),:PRINT " ..... 15"
1050 PRINT "          M23AB = ";A(16),:PRINT " ..... 16"
1060 PRINT "          M13BB = ";A(17),:PRINT " ..... 17"
1070 PRINT "          M23BB = ";A(18),:PRINT " ..... 18"
1080 PRINT :PRINT :PRINT TAB(20)" NUMERO A CORRIGER (0 = arrêt des corrections)
: ";:INPUT " ",Z
1090 IF Z<0 OR Z>18 THEN 1080
1100 IF Z=0 THEN OPEN "D",#1,F$:FOR I=1 TO 18:PRINT #1,A(I):NEXT I:CLOSE:GOTO 3:
0
1110 PRINT TAB(20)" ANCIENNE VALEUR : ";A(Z);:INPUT "      NOUVELLE VALEUR : ",A(
)
1120 GOTO 880
1130 '***** CALCUL DES PARAMETRES *****
1140 CLOSE :OPEN "I",#1,F$
1150 INPUT #1,N1A,N1B,N2A,M12AA,M12BA,N2B,M12AB,M12BB,N3A,M13AA,M23AA,M13BA,M23B
A,N3B,M13AB,M23AB,M13BB,M23BB
1160 CLOSE 1
1170 CLS
1180 PRINT :PRINT :PRINT
1190 PRINT TAB(30)"CHOIX DU MODELE"
1200 PRINT :PRINT
1210 PRINT TAB(15) "MODELE GENERAL ..... 1"
1220 PRINT TAB(15) "MODELE A MIGRATION PARTIELLE (MODELE D'IWAO) .... 2"
1230 PRINT TAB(15) "MODELE A MIGRATION RESTREINTE ..... 3"
1240 PRINT TAB(15) "RETOUR AU MENU GENERAL ..... 4"
1250 PRINT :PRINT :PRINT TAB(30);:INPUT "NUMERO CHOISI : ",I
1260 IF I<1 OR I>4 THEN 1250
1270 ON I GOTO 1280,1540,1780,350
1280 '
1290 '***** MODELE GENERAL *****
1300 '
1310 CLS
1320 DET12=M12AA*M12BB-M12BA*M12AB
1330 DET23=M23AA*M23BB-M23BA*M23AB
1340 '
1350 P1A=DET12/(N2A*M12BB-N2B*M12BA)
1360 NT1A=N1A/P1A
1370 PHI1AA=(M12AA/N1A)+N2A*(M13AA*M23BB-M13AB*M23BA)/(N1A*DET23)
1380 PHI1AB=(M12AB/N1A)+N2B*(M13AB*M23AA-M13AA*M23AB)/(N1A*DET23)
1390 PHI1A=PHI1AA+PHI1AB
1400 P2A=DET23/(N3A*M23BB-N3B*M23BA)
1410 NT2A=N2A/P2A
1420 PHI2AAP3A=M23AA/N2A
1430 PHI2BAP3A=M23BA/N2B
1440 P1B=DET12/(N2B*M12AA-N2A*M12AB)
1450 NT1B=N1B/P1B
1460 PHI1BB=(M12BB/N1B)+N2B*(M13BB*M23AA-M13BA*M23AB)/(N1B*DET23)
1470 PHI1BA=(M12BA/N1B)+N2A*(M13BA*M23BB-M13BB*M23BA)/(N1B*DET23)
1480 PHI1B=PHI1BB+PHI1BA
1490 P2B=DET23/(N3B*M23AA-N3A*M23AB)
1500 NT2B=N2B/P2B

```

```

1510 PHI2BBP3B = M23BB / N2B
1520 PHI2ABP3B = M23AB / N2A
1530 MODEL = 1 : GOTO 2080 : 'SORTIE DES RESULTATS
1540 '***** MODELE D'IWAO *****
1550 CLS
1560 DET12=M12AA*M12BB-M12BA*M12AB
1570 DET23=M23AA*M23BB-M23BA*M23AB
1580 '
1590 P1A=DET12/(N2A*M12BB-N2B*M12BA)
1600 NT1A=N1A/P1A
1610 PHI1AA=(M13AA*N2A+M12AA*M23AA) / (M23AA*N1A)
1620 PHI1BA = PHI1AA*N1A*M12BA / (N1B*M12BB)
1630 P2A = DET23/(N3A*M23BB-N3B*M23BA)
1640 NT2A = N2A/P2A
1650 P1B=DET12/(N2B*M12AA-N2A*M12AB)
1660 NT1B=N1B/P1B
1670 PHI1BB=(N2B*M13BB + M12BB*M23BB) / (M23BB*N1B)
1680 PHI1AB = PHI1BB*N1B*M12AB / (N1A*M12AA)
1690 PHI1A = PHI1AA+PHI1AB
1700 PHI1B = PHI1BB+PHI1BA
1710 P2B = DET23/(N3B*M23AA-N3A*M23AB)
1720 NT2B = N2B/P2B
1730 PHI2AAP3A = M23AA / N2A
1740 PHI2BAP3A = M23BA * P1B / M12BB
1750 PHI2BBP3B = M23BB / N2B
1760 PHI2ABP3B = M23AB * P1A / M12AA
1770 MODEL = 2 : GOTO 2080
1780 '+++++++ MIGRATION RESTREINTE .....
1790 CLS
1800 DET12=M12AA*M12BB-M12BA*M12AB
1810 DET23=M23AA*M23BB-M23BA*M23AB
1820 '
1830 P1A=DET12/(N2A*M12BB-N2B*M12BA)
1840 NT1A=N1A/P1A
1850 PHI1AA=(M13AA*N2A+M12AA*M23AA) / (M23AA*N1A)
1860 P2A = DET23/(N3A*M23BB-N3B*M23BA)
1870 NT2A = N2A/P2A
1880 P1B=DET12/(N2B*M12AA-N2A*M12AB)
1890 NT1B=N1B/P1B
1900 PHI1BB=(N2B*M13BB + M12BB*M23BB) / (M23BB*N1B)
1910 P2B = DET23/(N3B*M23AA-N3A*M23AB)
1920 NT2B = N2B/P2B
1930 PHI2AAP3A = M23AA / N2A
1940 PHI2BBP3B = M23BB / N2B
1950 MODEL = 3 : GOTO 2080
1960 '+++++++ SOUS PROGRAMME DE DEMANDE DE FICHER *****
1970 CLOSE
1980 IF F# ="" THEN 1990 ELSE 2020
1990 PRINT : INPUT "NOM DU FICHER (Avec le chemin d'accès complet) : ",F#
2000 IF F# ="" THEN 1990

```

```

2010 F$=F$+".MIG":RETURN
2020 IF CREAT = 1 THEN GOTO 1990
2030 PRINT :PRINT "LE FICHER PAR DEFAULT EST : ";F$
2040 INPUT "EST CE VOTRE FICHER DE TRAVAIL (O/N) : ",R$
2050 IF (R$<>"o" AND R$<>"O" AND R$<>"n" AND R$<>"N") THEN 2040
2060 IF R$="O" OR R$="o" THEN RETURN
2070 GOTO 1990
2080 '+++++ SORTIE DES RESULTATS +++++
2090 A$="N"
2100 OPEN "SCRN:" FOR OUTPUT AS #1
2110 IF A$="N" OR A$="n" THEN 2170
2120 PRINT #1,TAB(20)"+++++"
2130 PRINT #1,TAB(20)"+ PARAMETRES DE DEUX POPULATIONS +"
2140 PRINT #1,TAB(20)"+ SUJETTES A MIGRATIONS ENTRE 2 ZONES +"
2150 PRINT #1,TAB(20)"+++++"
2160 PRINT #1,"":PRINT #1,"FICHER : ",F$:PRINT #1,"":PRINT #1,""
2170 ON MODEL GOTO 2180,2190,2200
2180 PRINT #1,TAB(30)"MODELE GENERAL":GOTO 2210
2190 PRINT #1,TAB(15)"MODELE A MIGRATION PARTIELLE (MODELE D'IWAD)":GOTO 2210
2200 PRINT #1,TAB(25)"MODELE A MIGRATION RESTREINTE"
2210 PRINT #1,""
2220 PRINT #1,"PHI(j,x,y) = probabilité qu'un animal, vivant et dans la zone x a
u temps j, soit vivant et dans la zone y au temps j+1"
2230 PRINT #1," PHI( 1,A,A ) = ";PHI1AA,
2240 IF MODEL = 3 THEN PRINT #1,"PHI( 1,A,B ) = NON CALCULABLE",:GOTO 2260
2250 PRINT #1,"PHI( 1,A,B ) = ";PHI1AB
2260 PRINT #1," PHI( 1,B,B ) = ";PHI1BB,
2270 IF MODEL = 3 THEN PRINT #1,"PHI( 1,B,A ) = NON CALCULABLE":GOTO 2290
2280 PRINT #1,"PHI( 1,B,A ) = ";PHI1BA
2290 PRINT #1,""
2300 PRINT #1,"PHI(j,x) = probabilité qu'un animal, vivant et dans la zone x au
temps j, soit vivant au temps j+1 [ = PHI(j,x,A) + PHI(j,
x,B) ]
2310 IF MODEL = 3 THEN PRINT #1," PHI( 1,A ) et PHI( 1,B ) sont NON CALC
ULABLES" :GOTO 2340
2320 PRINT #1," PHI( 1,A ) = ";PHI1A,
2330 PRINT #1,"PHI( 1,B ) = ";PHI1B
2340 PRINT #1,""
2350 PRINT #1,"P ( j,x ) = probabilité qu'un animal, vivant et dans la zone x au
temps j, soit capturé dans l'échantillon pris au temps j.
2360 PRINT #1," P ( 1,A ) = ";P1A,
2370 PRINT #1,"P ( 1,B ) = ";P1B
2380 PRINT #1," P ( 2,A ) = ";P2A,
2390 PRINT #1,"P ( 2,B ) = ";P2B
2400 PRINT #1,""
2410 PRINT #1,"N ( j,x ) = taille de la population vivant dans la zone x au temps
j"
2420 PRINT #1," N ( 1,A ) = ";NT1A,
2430 PRINT #1,"N ( 1,B ) = ";NT1B
2440 PRINT #1," N ( 2,A ) = ";NT2A,
2450 PRINT #1,"N ( 2,B ) = ";NT2B
2460 PRINT #1,""
2470 IF MODEL = 3 THEN PRINT #1,"PHI(2,A,A)*P(3,A) = ";PHI2AAP3A,:PRINT #1,"PHI(
2,B,A)*P(3,A) = NON CALCULABLE",
2480 IF MODEL = 3 THEN PRINT #1,"PHI(2,B,B)*P(3,B) = ";PHI2BBP3B,:PRINT #1,"PHI(
2,A,B)*P(3,B) = NON CALCULABLE":GOTO 2510
2490 PRINT #1,"PHI(2,A,A)*P(3,A) = ";PHI2AAP3A,:PRINT #1,"PHI(2,B,A)*P(3,A) = ";
PHI2BAP3A,
2500 PRINT #1,"PHI(2,B,B)*P(3,B) = ";PHI2BBP3B,:PRINT #1,"PHI(2,A,B)*P(3,B) = ";
PHI2ABP3B

```

```

2510 PRINT #1, ""
2520 INPUT "Impression sur imprimante (O/N) : ", A$
2530 IF A$ <> "O" AND A$ <> "o" AND A$ <> "N" AND A$ <> "n" THEN 2520
2540 IF (A$ = "o" OR A$ = "O") THEN CLOSE : OPEN "LPT1:" FOR OUTPUT AS #1 : GOTO 2110
2550 CLOSE : GOTO 1130
2560 '+++++++ SOUS PROGRAMME D'EDITION +++++++
2570 IF A$ = "N" OR A$ = "n" THEN OPEN "SCRN:" FOR OUTPUT AS #1 ELSE OPEN "LPT1:" F
R OUTPUT AS #1
2580 PRINT #1, "": PRINT #1, ""
2590 IF A$ = "N" OR A$ = "n" THEN 2650
2600 PRINT #1, TAB(20) "+++++++ "
2610 PRINT #1, TAB(20) " + PARAMETRES DE DEUX POPULATIONS + "
2620 PRINT #1, TAB(20) " + SUJETTES A MIGRATIONS ENTRE 2 ZONES + "
2630 PRINT #1, TAB(20) "+++++++ "
2640 PRINT #1, ""
2650 PRINT #1, "NOM DU FICHER : ", F$
2660 PRINT #1, "": PRINT #1, ""
2670 PRINT #1, "PREMIER ECHANTILLONNAGE ( Temps t = 1 )"
2680 PRINT #1, ""
2690 PRINT #1, " TOTAL CAPTURE ET MARQUE DANS LA ZONE A : "; GOSUB 3030
2700 PRINT #1, " TOTAL CAPTURE ET MARQUE DANS LA ZONE B : "; GOSUB 3030
2710 PRINT #1, "": PRINT #1, ""
2720 PRINT #1, "DEUXIEME ECHANTILLONNAGE ( Temps t = 2 )"
2730 PRINT #1, ""
2740 PRINT #1, " ZONE A : "
2750 PRINT #1, " TOTAL CAPTURE EN A : "; : GOSUB 3030
2760 PRINT #1, " NBRE MARQUES EN A (à t = 1) ET RECAPTURES EN A : "; : GO
UB 3030
2770 PRINT #1, " NBRE MARQUES EN B (à t = 1) ET RECAPTURES EN A : "; : G
SUB 3030
2780 PRINT #1, ""
2790 PRINT #1, " ZONE B : "
2800 PRINT #1, " TOTAL CAPTURE EN B : "; : GOSUB 3030
2810 PRINT #1, " NBRE MARQUES EN A (à t = 1) ET RECAPTURES EN B : "; : G
SUB 3030
2820 PRINT #1, " NBRE MARQUES EN B (à t = 1) ET RECAPTURES EN B : "; : G
SUB 3030
2830 IF A$ = "O" OR A$ = "o" THEN 2860
2840 IF CREAT = 1 THEN 2860
2850 PRINT : INPUT "Faites un 'retour chariot' pour la suite ...", R$
2860 PRINT #1, "": PRINT #1, ""
2870 PRINT #1, "TROISIEME ECHANTILLONNAGE"
2880 PRINT #1, ""

```

```

2890 PRINT #1,"
2900 PRINT #1,"
2910 PRINT #1,"
      JSUB 3030
2920 PRINT #1,"
      JSUB 3030
2930 PRINT #1,"
      JSUB 3030
2940 PRINT #1,"
      JSUB 3030
2950 PRINT #1,""
2960 PRINT #1,"
      ZONE B : "
2970 PRINT #1,"
      TOTAL CAPTURE EN B : ";: GOSUB 3030
2980 PRINT #1,"
      NBRE MARQUES EN A (à t = 1) ET RECAPTURES EN B : "; : G
      JSUB 3030
2990 PRINT #1,"
      NBRE MARQUES EN A (à t = 2) ET RECAPTURES EN B : "; : G
      JSUB 3030
3000 PRINT #1,"
      NBRE MARQUES EN B (à t = 1) ET RECAPTURES EN B : "; : G
      JSUB 3030
3010 PRINT #1,"
      NBRE MARQUES EN B (à t = 2) ET RECAPTURES EN B : "; : G
      JSUB 3030
3020 RETURN
3030 '+++++
3040 IF CREAT = 1 THEN INPUT "",A(K) ELSE PRINT #1,A(K)
3050 K = K + 1
3060 RETURN
3070 '+++++

```